Maple-assisted proof of formula for A295601

Robert Israel

24 November 2017

There are $2^6 = 64$ possible configurations for a 2×3 sub-array. Consider the 64×64 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×3 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j, and each 1 in that row is horizontally or vertically adjacent to 0, 2, 3 or 4 1's), and 0 otherwise. The following Maple code computes it. I'm encoding a configuration

$$\begin{bmatrix} b_1 & b_2 & b_3 \\ b_4 & b_5 & b_6 \end{bmatrix}$$

as $b+1$ where $b_1 b_2 b_3 b_4 b_5 b_6$ is the binary representation of b. The $+1$ is needed because matrix indices start at 1 rather than 0.

```
> q:= proc(a,b) local r,s,t,M,i;
    s:= floor((a-1)/8);
    if s <> (b-1) mod 8 then return 0 fi;
    s:= convert(s+8,base,2);
    r:= convert(8+floor((b-1)/8),base,2);
    t:= convert(8+ ((a-1) mod 8),base,2);
    M:= Vector(3);
    if s[1] = 1 and s[2] = 1 then M[1]:= 1; M[2]:= 1 fi;
    if s[2]=1 and s[3]=1 then M[2]:= M[2]+1; M[3]:= 1 fi;
    for i from 1 to 3 do if s[i]=1 then
      M[i]:= M[i]+r[i]+t[i];
      if M[i] = 1 then return 0 fi;
    fi od;
    1
  end proc:
  T:= Matrix(64,64, q);
```

$$T := \begin{bmatrix} \textit{64 x 64 Matrix} \\ \textit{Data Type: anything} \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{bmatrix} \qquad \textbf{(1)}$$

Thus $a(n) = u\,T^n\,v$ where u and v are row and column vectors respectively with $u_i = 1$ for i corresponding to configurations with bottom row $(0, 0, 0)$, 0 otherwise, and $v_i = 1$ for i corresponding to configurations with top row $(0, 0, 0)$, 0 otherwise. The following Maple code produces these vectors.

```
> u:= Vector[row](64):
  v:= Vector(64):
  for i from 0 to 7 do u[8*i+1]:= 1; v[i+1]:= 1;
  od:
```

To check, here are the first few entries of our sequence.

```
> seq(u . T^n . v, n = 1 .. 10);
```

$$5, 20, 83, 338, 1425, 6080, 26249, 114298, 501405, 2211832 \quad (2)$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \to t^{21} - 8\,t^{20} + 17\,t^{19} - 4\,t^{18} - 18\,t^{17} + 54\,t^{16} - 48\,t^{15} + 118\,t^{14} + 8\,t^{13} + 46\,t^{12} + 95\,t^{11} - 462\,t^{10} + 39\,t^{9} - 188\,t^{8} - 421\,t^{7} + 376\,t^{6} + 182\,t^{5} - 112\,t^{4} - 23\,t^{3} + 12\,t^{2} \quad (3)$$

This turns out to have degree 21. Thus we will have $0 = u\,P(T)\,T^n\,v = \sum_{i=0}^{21} p_i\,a(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 21, which would hold true for any u and v. It seems that with our particular u and v we have a recurrence of order only 12, corresponding to a factor of P.

```
> factor(P(t));
```

$$t^2\,(t+1)\,(t^{12} - 7\,t^{11} + 9\,t^{10} + 11\,t^{9} - 14\,t^{8} + 53\,t^{7} - 19\,t^{6} - 42\,t^{5} + 23\,t^{4} - 72\,t^{3} - 36\,t^{2} + 25\,t + 12)\,(t^6 - 2\,t^5 + 3\,t^4 - 2\,t^3 + 7\,t^2 - 5\,t + 1) \quad (4)$$

```
> Q:= unapply(t^12-7*t^11+9*t^10+11*t^9-14*t^8+53*t^7-19*t^6-42*
  t^5+23*t^4-72*t^3-36*t^2+25*t+12, t);
```

$$Q := t \to t^{12} - 7\,t^{11} + 9\,t^{10} + 11\,t^{9} - 14\,t^{8} + 53\,t^{7} - 19\,t^{6} - 42\,t^{5} + 23\,t^{4} - 72\,t^{3} - 36\,t^{2} + 25\,t + 12 \quad (5)$$

The complementary factor $R(t) = \frac{P(t)}{Q(t)}$ has degree 9.

```
> R:= unapply(normal(P(t)/Q(t)), t);
```

$$R := t \to (t^7 - t^6 + t^5 + t^4 + 5\,t^3 + 2\,t^2 - 4\,t + 1)\,t^2 \quad (6)$$

Now we want to show that $b(n) = u\,Q(T)\,T^n\,v = 0$ for all n. This will certainly satisfy the order-5 recurrence

$$\sum_{i=0}^{9} r_i\,b(i+n) = \sum_{i=0}^{9} r_i u\,Q(T)\,T^{n+i} v = u\,Q(T)\,R(T)\,T^n\,v = u\,P(T)\,T^n\,v = 0$$

where r_i are the coefficients of $R(t)$. To show all $b(n) = 0$ it suffices to show $b(0) = ... = b(8) = 0$.

```
> seq( u . Q(T) . T^n . v, n = 0 .. 8);
```

$$0, 0, 0, 0, 0, 0, 0, 0, 0 \quad (7)$$